Лекция 8

Функции от по-висок ред в Haskell

Шаблони на пресмятания със списъци (Patterns of Computation over Lists)

Голяма част от функциите за работа със списъци, които разгледахме досега, могат да бъдат отнесени към малък брой типове. При това за съставянето на тези типове функции могат да бъдат използвани съответни ***шаблони на пресмятания*** (***patterns of computation***).

Примери за шаблони на пресмятания (програмни шаблони)

- Прилагане на една и съща операция върху всички елементи на даден списък (***mapping***)
  Примери:
  - Реализацията на хоризонталното завъртане flipH (в системата за манипулиране на картинки)
  - Извличането на вторите елементи на двойките от даден списък (при работата с библиотечната “база от данни”)
  - Конвертирането на списъка от бар кодове в списък от вида (Name, Price) (в примера за конструиране на касова бележка)

- Извличане на елементите на даден списък, които удовлетворяват дадено условие (***filtering***)
  Примери:
  - Извличането на тези двойки, първият елемент на които съвпада с дадено име (при работата с библиотечната “база от данни”)
  - Извличането на цифрите/буквите от даден символен низ

- Комбиниране / акумулиране на елементите на даден списък (***folding***)
  Примери:
  - Конкатенацията на елементите на даден списък от списъци
  - Събирането / умножаването на елементите на даден списък от числа

Реализацията на тези и други шаблони на пресмятания (patterns of computation) може да се извърши с помощта на подходящи ***функции от по-висок ред***.

Дефиниция

Функция от по-висок ред се нарича всяка функция, която получава поне една функция като параметър (аргумент) или връща функция като резултат.

Наличието на средства за дефиниране и използване на функции от по-висок ред съществено увеличава изразителната сила на съответния език за програмиране.

Функциите като параметри

Тук ще покажем как могат да се дефинират някои често използвани функции от по-висок ред за работа със списъци. Тези функции са дефинирани в Prelude.hs, т.е. нашите дефиниции ще бъдат направени само с учебна цел.

Прилагане на дадена функция към всички елементи на даден списък (map)

Декларация на типа:

```
map :: (a -> b) -> [a] -> [b]
```



Дефиниция (първи вариант):

```
map f xs = [f x| x <- xs]
```

Дефиниция (втори вариант):

```
map f []     = []
map f (x:xs) = f x : map f xs
```

Примери

```
doubleAll :: [Int] -> [Int]
-- Удвоява елементите на даден списък.
doubleAll xs = map double xs
  where
    double :: Int -> Int
    double x = 2*x

convertChrs :: [Char] -> [Int]
-- Конвертира знаковете от даден списък
-- в техните ASCII кодове.
convertChrs xs = map ord xs
```

```
type Picture = [String]
flipH :: Picture -> Picture
-- Завърта дадена картинка около ординатната ос
-- (“завърта” хоризонталните координати на точките
-- от картинката).
flipH pic = map reverse pic
```

Филтриране на елементите на даден списък (filter)

Декларация на типа:

```
filter :: (a -> Bool) -> [a] -> [a]
```



Дефиниция (първи вариант):

```
filter p xs = [x | x <- xs, p x]
```

Дефиниция (втори вариант):

```
filter p []     = []
filter p (x:xs)
   | p x        = x : filter p xs
   | otherwise =     filter p xs
```

Примери

```
isEven :: Int -> Bool
isEven n = (n `mod` 2 == 0)

isSorted :: [Int] -> Bool
isSorted xs = (xs == iSort xs)

filter isEven [2,3,4,5] ⟶ [2,4]
filter isSorted [[2,3,4,5],[3,2,5],[],[3]]
 ⟶ [[2,3,4,5],[],[3]]
```

Комбиниране на zip и map (zipWith)

Вече разгледахме полиморфичната функция **`zip :: [a] -> [b] -> [(a,b)]`**, която комбинира два дадени списъка в списък от двойки от съответните елементи на тези списъци.

Ще дефинираме нова функция, zipWith, която комбинира ефекта от действието на zip и map.

Декларация на типа

Функцията zipWith ще има три аргумента: една функция и два списъка. Двата списъка (вторият и третият аргумент на zipWith) са от произволни типове (съответно [a] и [b]). Резултатът също е списък от произволен тип ([c]). Първият аргумент на zipWith е функция, която се прилага върху аргументи – съответните елементи на двата списъка и връща съответния елемент на резултата, т.е. това е функция от тип **`a -> b -> c.`**

Следователно

```
zipWith :: (a -> b -> c) -> [a] -> [b] -> [c]
```

Дефиниция

```
zipWith f (x:xs) (y:ys) = f x y : zipWith f xs ys
zipWith f  _      _     = []
```

Примери

Нека plus и mult са функции, дефинирани както следва:

```
plus :: Int -> Int -> Int
plus a b = a+b

mult :: Int -> Int -> Int
mult a b = a*b
```

Тогава

```
zipWith plus [1,2,3]      [4,5,6] ⟶ [5,7,9]
zipWith mult [1,2,3,4,5] [6,7,8] ⟶ [6,14,24]
```

Нека се върнем още един път на примерната система за манипулиране на картинки. В нея дефинирахме функцията

```
sideBySide :: Picture -> Picture -> Picture
sideBySide p q = [pline ++ qline
     | (pline,qline) <- zip p q]
```

Нова дефиниция на функцията sideByside:

```
sideBySide pic1 pic2 = zipWith (++) pic1 pic2
```

Забележка. Функциите map, filter и zipWith са дефинирани в Prelude.hs.

Комбиниране/акумулиране на елементите на даден списък (foldr1 и foldr)

Тук ще разгледаме група функции от по-висок ред, които реализират операцията *комбиниране* (*акумулиране*) на елементите на даден списък, използвайки подходяща функция.

Тази операция е достатъчно обща и се реализира от множество стандартни функции, включени в Prelude.hs.

Действие на функцията foldr1

Дефиницията на тази функция включва два случая:

- foldr1, приложена върху дадена функция f и списък от един елемент [a], връща като резултат a;
- Прилагането на foldr1 върху функция и по-дълъг списък е еквивалентно на

  foldr1 f [e_1,e_2, … , e_k]
  = e_1 `f` (e_2 `f` (... `f` e_k) ...)
  = e_1 `f` (foldr1 f [e_2, … , e_k])
  = f e_1 (foldr1 f [e_2, ... , e_k])

Съответната дефиниция на Haskell изглежда както следва:

```
foldr1 :: (a -> a -> a) -> [a] -> a
foldr1 f [x]    = x
foldr1 f (x:xs) = f x (foldr1 f xs)
```

Примери

```
foldr1 (+) [3,98,1] = 102
foldr1 (||) [False,True,False] = True
foldr1 min [6] = 6
foldr1 (*) [1 .. 6] = 720
```

Забележка. Функцията foldr1 не е дефинирана върху втори аргумент – празен списък.

Разглежданата функция може да бъде модифицирана така, че да получава един допълнителен аргумент, който определя стойността, която следва да се върне при опит за комбиниране по зададеното правило на елементите на празния списък.

Новополучената функция се нарича ***foldr*** (което означава ***fold***, т.е. комбиниране/акумулиране, извършено чрез групиране от дясно – bracketing to the **r**ight).

Дефиниция на функцията foldr:

```
foldr :: (a -> a -> a) -> a -> [a] -> a
```




```
foldr f s []     = s
foldr f s (x:xs) = f x (foldr f s xs)
```

Примери

```
concat :: [[a]] -> [a]
concat xs = foldr (++) [] xs

and :: [Bool] -> Bool
and bs = foldr (&&) True bs
```

Забележка. В действителност типът на функцията foldr е по-общ:

```
foldr :: (a -> b -> b) -> b -> [a] -> b
```

Функциите като върнати стойности

Дефиниции на функции на функционално ниво

Дефинирането на някаква функция на функционално ниво предполага действието на тази функция да се опише не в термините на резултата, който връща тя при прилагане към подходящо множество от аргументи, а като директно се посочи връзката ѝ с други функции.

Например, ако вече са дефинирани функциите

`f :: b -> c` и

`g :: a -> b` , то

тяхната композиция f . g (т.е. функцията, за която е изпълнено (f . g) x = f (g x) за всяко x от тип a) може да се дефинира с използване на вградения оператор '.' от тип

`(.) :: (b -> c) -> (a -> b) -> (a -> c)` .

Пример 1. Двукратно прилагане на функция.

```
twice :: (a -> a) -> (a -> a)
twice f = (f . f)
```

Нека например succ е функция, която прибавя 1 към дадено цяло число:

```
succ :: Int -> Int
succ n = n + 1
```

Тогава

```
(twice succ) 12
 ⟶ (succ . succ) 12
 ⟶ succ (succ 12)
 ⟶ 14
```

Пример 2. n-кратно прилагане на функция.

```
iter :: Int -> (a -> a) -> (a -> a)
iter n f
   | n>0       = f . iter (n-1) f
   | otherwise = id
```

Тук id е вградената функция – идентитет.

Дефиниране на функция, която връща функция като резултат

Пример. Функция, която за дадено цяло число n връща като резултат функция на един аргумент, която прибавя n към аргумента си.

```
addNum :: Int -> (Int -> Int)
addNum n = addN
           where
           addN m = n+m
```

Така оценка на обръщението към функцията addNum ще бъде функцията с име addN. От своя страна функцията addN е дефинирана в клаузата where.

Използваният подход може да бъде окачествен като индиректен: най-напред посочваме името на функцията – резултат и едва след това дефинираме тази функция.

Ламбда нотация
(ламбда изрази)

Вместо да именуваме и да дефинираме някаква функция, която бихме искали да използваме, можем да запишем директно тази функция.

Например в случая на дефиницията на addNum резултатът може да бъде дефиниран като

```
\m -> n+m
```

В Haskell изрази от този вид се наричат ***ламбда изрази***, а функциите, дефинирани чрез ламбда изрази, се наричат *анонимни функции*.

Забележка. Символът “\” е избран за означаване на ламбда изразите, защото той наподобява на гръцката буква **λ**.

Дефиницията на функцията addNum с използване на ламбда израз придобива вида

```
addNum n = (\m -> n+m)
```

Частично прилагане на функции

Нека разгледаме като пример функцията за умножение на две числа, дефинирана както следва:

```
multiply :: Int -> Int -> Int
multiply x y = x*y
```

Ако тази функция бъде приложена към два аргумента, като резултат ще се получи число, например multiply 2 3 връща резултат 6.

Какво ще се случи, ако multiply се приложи към един аргумент, например числото 2?

Отговорът е, че като резултат ще се получи функция на един аргумент y, която удвоява аргумента си, т.е. връща резултат 2*y.

Следователно, **всяка функция на два или повече аргумента може да бъде приложена частично към по-малък брой аргументи**. Тази идея дава богати възможности за конструиране на функции като оценки на обръщения към други функции.

Пример. Функцията doubleAll, която удвоява всички елементи на даден списък от цели числа, може да бъде дефинирана както следва:

```
doubleAll :: [Int] -> [Int]
doubleAll = map (multiply 2)
```

Тип на резултата от частично прилагане на функция

Правило на изключването

Ако дадена функция f е от тип
$\mathtt{t_1 \text{ -> } t_2 \text{ -> } \ldots \text{ -> } t_n \text{ -> } t}$
и тази функция е приложена към аргументи
$\mathtt{e_1 :: t_1, e_2 :: t_2, \ldots , e_k :: t_k}$ (където $k \leq n$),
то типът на резултата се определя чрез изключване на типовете t_1, t_2, … , t_k:
$\mathtt{\cancel{t_1} \text{ -> } \cancel{t_2} \text{ -> } \ldots \text{ -> } \cancel{t_k} \text{ -> } t_{k+1} \text{ -> } \ldots \text{ -> } t_n \text{ -> } t,}$
т.е. резултатът е от тип
$\mathtt{t_{k+1} \text{ -> } t_{k+2} \text{ -> } \ldots \text{ -> } t_n \text{ -> } t.}$

Примери

```
multiply 2 :: Int -> Int
multiply 2 3 :: Int

doubleAll :: [Int] -> [Int]
doubleAll [2,3] :: [Int]
```

Забележки

1. Прилагането на функция е ***ляво асоциативна операция***, т.е.

```
f x y = (f x) y    и
f x y ≠ f (x y)
```

2. Операторът ‘**->**’ не е асоциативен.

   Например записите

```
f :: Int -> Int -> Int    и
g :: (Int -> Int) -> Int
```

означават функции от различни типове.

Частичното прилагане на функции налага нова гледна точка върху понятието “брой на аргументите на дадена функция”. От тази гледна точка може да се каже, че всички функции в Haskell имат по един аргумент. Ако резултатът от прилагането на функцията върху даден аргумент е функция, тази функция може отново да бъде приложена върху един аргумент и т.н.

Сечения на оператори
(operator sections)

Операторите в Haskell могат да бъдат прилагани частично, като за целта се задава това, което е известно, под формата на т. нар. ***сечения на оператори*** (***operator sections***).

Примери

(+2) Функцията, която прибавя към аргумента си числото 2.
(2+) Функцията, която прибавя аргумента си към числото 2.
(>2) Функцията, която проверява дали дадено число е по-голямо от 2.
(3:) Функцията, която поставя числото 3 в началото на даден списък.

(++”\n”) Функцията, която поставя *newline* в края на даден низ.

(“\n”++) Функцията, която поставя *newline* в началото на даден низ.

Общото правило гласи, че сечението на оператора ***op*** “добавя” аргумента си по начин, който завършва от синтактична гледна точка записа на приложението на оператора (обръщението към оператора).

С други думи,
(op x) y = y op x
(x op) y = x op y

Когато бъде комбинирана с функции от по-висок ред, нотацията на сечението на оператори е едновременно мощна и елегантна. Тя позволява да се дефинират разнообразни функции от по-висок ред.

Например,
filter (>0) . map (+1)

е функцията, която прибавя 1 към всеки от елементите на даден списък, след което премахва тези елементи на получения списък, които не са положителни числа.